
付録 C 
見積もりアプリケーションのサンプルコード 

 
この付録では、12 ページの「Apex コード入門」で説明した見積もりアプリケーションの全サンプ

ルコードを示します。 

 

Note： この文書のコード例は、リテラル文字列（引用符で囲まれているテキスト）の一部

に改行が挿入され読みやすくなっていますが、Apex Code スクリプトでは、リテラル文字列

で改行することはできません。 この文書のコード例をそのままコピーすると、エラーにな

る可能性があります。 
 
oppLineTrigger.apex 

 
// 商談商品がデータベースに保存される前に 
// カスタム項目 color__c の値を設定するトリガ 
trigger oppLineTrigger on OpportunityLineItem (before insert) { 
 

// 重複のない価格表エントリを取得する 
Set<Id> pbeIds = new Set<Id>(); 
for (OpportunityLineItem oli : Trigger.new) 

pbeIds.add(oli.pricebookentryid); 
 

// それらの価格表エントリを照会する 
Map<Id, PricebookEntry> entries = new Map<Id, PricebookEntry>( 

[select product2.color__c from pricebookentry 
where id in :pbeIds]); 
 

// 商談商品の色を設定する 
for (OpportunityLineItem oli : Trigger.new) 

oli.color__c = entries.get(oli.pricebookEntryId).product2.color__c; 
} 

 

oppTrigger.apex 

 
// 見積もりを持つ商談の削除を抑止するトリガ 
trigger oppTrigger on Opportunity (before delete) { 
 

// Trigger.oldMap は、削除されようとしている商談に ID から 
// マップするトリガコンテキスト変数。 
// これを含む SOQL クエリを発行し、関連する見積もりのリストを作る 
for (Quote__c q : [select opportunity__c from quote__c 

where opportunity__c in :Trigger.oldMap.keySet()]) { 
// 有効な見積もりごとに Trigger.oldMap から関連する商談を取り出し、 
// その商談にエラーメッセージを付けることにより 
// 削除を抑止する 

Trigger.oldMap.get(q.opportunity__c).addError( 
'Cannot delete opportunity with a quote'); 

 } 



 200

} 

 

quoteLineTrigger.apex 

 
// このトリガは、見積もり品目が変更されたときに、 
// その見積もりの合計金額を更新する 
trigger quoteLineTrigger on quoteLine__c 
 

(before insert, before update, 
after insert, after update, after delete) { 
// Trigger.isBefore や Trigger.isUpdate などのトリガコンテキスト変数を利用すれば、 
// 起動された条件により実行する Apex コードを選ぶ多条件トリガを作ることができる 
if (Trigger.isBefore) { 

for (quoteLine__c ql : Trigger.new) { 
 

// 見積もり品目が更新または挿入される前にこのトリガが起動された場合は、 
// 各品目の TotalPrice__c 項目が再計算される。 
// なお、この計算は、 
// このアプリケーションの計算項目で実装することもできる 
if (Trigger.isUpdate || Trigger.isInsert) { 

ql.totalprice__c = ql.unitprice__c * ql.quantity__c; 
} 

} 
} else { 
 

// 課題 1 と同様に、 
// 集合を使って変更中の見積もりすべての重複のないリストを作る。 
// これは、定型一括処理の第 1 ステップ 
Set<Id> quoteIds = new Set<Id>(); 
 
// ここで、リストの大きさを取得する際と見積もりの ID を取得する際に、 
// 三項演算子を使っている点に注意 
// 削除の場合は Trigger.old リスト変数、 
// 挿入と更新の場合は Trigger.new リスト変数を用いる 
for (Integer i = 0; i < (Trigger.isDelete ? Trigger.old.size() 

: Trigger.new.size()); i++) { 
quoteIds.add(Trigger.isDelete ? Trigger.old[i].quote__c 

: Trigger.new[i].quote__c); 
} 
 
// 定型一括処理の第 2 ステップ 
// この集合を含む SOQL クエリを発行し、Quote に対する ID のマップを作る 
Map<Id, Quote__c> quotes = new Map<Id, Quote__c>( 

[select id, total__c from quote__c where id in :quoteIds]); 
 

// Trigger.old または Trigger.new に含まれる各見積もり品目の quoteId を取得する 
for (Integer i = 0; i < (Trigger.isDelete ? Trigger.old.size() 

: Trigger.new.size()); i++) { 
String quoteId = Trigger.isDelete ? Trigger.old[i].quote__c 

: Trigger.new[i].quote__c; 
 

// 見積もり品目が更新または削除された後に起動された場合は、 
// 親見積もりの total__c 項目を再計算する必要がある。 
// 見積もりの挿入中に起動された場合 

// （つまり、util.copyLineItemsFromOpp メソッドと同じ見積もりを 

// 対象としている場合）は、以降の処理をスキップする（課題 2 を参照） 
if (util.inQuoteTrigger) continue; 
 
// 定型一括処理の第 3 ステップ 

// 親見積もりの取得（上で作った集合の中にあるはず） 
quote__c q = quotes.get(quoteId); 
// 見積総額が未設定の場合は、 
// ここでゼロに設定しておく 
if (q.total__c == null) q.total__c = 0; 
 
// 見積もり品目が挿入された場合は、 
// 見積もりの総額に加算 
if (Trigger.isInsert && Trigger.new[i].totalprice__c != null) { 

q.total__c += Trigger.new[i].totalprice__c; 
} 
 

// 見積もり品目が更新された場合は、 

// 新しい品目の総額を加算し、古い品目の総額を減算 
if (Trigger.isUpdate) { 

if (Trigger.new[i].totalprice__c != null) 
q.total__c += Trigger.new[i].totalprice__c; 

if (Trigger.old[i].totalprice__c != null) 
q.total__c -= Trigger.old[i].totalprice__c; 

} 
 

// 見積もり品目が削除された場合は、減算 
if (Trigger.isDelete && Trigger.old[i].totalprice__c != null) { 

q.total__c -= Trigger.old[i].totalprice__c; 
} 

} 
// 定型一括処理の第 4 （最終）ステップ 
// すべての見積もりをまとめて更新する 
update quotes.values(); 

  } 
} 

 

quoteTrigger.apex 

 
// 商談の主見積もりの削除と、既存見積もりの Primary 項目のリセットを抑止する見積もりトリガ 
// このトリガには、課題 2 のコードも含まれている点に注意 
trigger quoteTrigger on quote__c (after insert, after update, 

before update, before delete) { 
if (Trigger.isBefore) { 

if (Trigger.isUpdate) { 
 

// まず、この更新が、主見積もりの Primary 項目を 

// 偽に設定するものではないことを確認する。 
// 更新されようとしている見積もりに対して、 
// 古いバージョンと新しいバージョンのコピーを作る 
// （新しいリストと古いリストを同時に使うのは、定型的な一括トリガ技法の一つ） 
for (Integer i = 0; i < Trigger.old.size(); i++) { 

Quote__c old = Trigger.old[i]; 
Quote__c nw = Trigger.new[i]; 

if (old.primary__c && !nw.primary__c) { 
 

// 見積もりの Primary フラグを付け替えるトランザクションの中を除き、 
// Primary フラグを別の見積もりに付け替えることはできない。 

// 次の if 文は 
// 古い見積もりの opportunityId と util クラスの opportunityId 変数を 

// 比較することで、上記条件の不成立を確認している 
// （課題 5 を参照） 
if (!util.opportunityIds.contains (old.opportunity__c)) { 
nw.Primary__c.addError('Primary quote cannot 

be marked non-primary'); 
} 

} 
} 

} 
// 次に、削除されようとしている見積もりが主見積もりでないことを確認する 
if (Trigger.isDelete) { 

for (Quote__c q : Trigger.old) { 
// 主見積もりを削除しようとしている場合は、 
// addError() を使って削除を抑止する 
if (q.primary__c) { 

q.addError('Primary quote cannot be deleted'); 
} 

} 
} 

} else { 
// 新しい見積もりの作成中に、opp lineitems をコピーする 
if (Trigger.isInsert) util.copyLineItemsFromOpp(Trigger.new); 
 
// 見積もりが挿入または更新されても、  

// Primary とされるレコードが 1 つだけであることを保証するために 
// Trigger.new に含まれ Primary になるはずの見積もりが参照する商談をすべて取得し、 
// それ以外の見積もりの Primary フラグを偽に設定する。 
// ただし、複数のレコードを同じ商談の主見積もりにしようとしている場合があり得るので 
// 注意が必要 
List<Quote__c> primaries = new List<Quote__c>(); 
 
// Primary にしようとしている見積もりを持つ商談の ID のマップの置き場所 
Map<Id, Id> opWithPrimary = new Map<Id, Id>(); 
 
// 複数の見積もりを同じ商談の主見積もりにしようとしている場合、その見積もりの置き場所 
Set<Id> badQuotes = new Set<Id>(); 
 
for (Integer i = 0; i < Trigger.new.size(); i++) { 

Quote__c nw = Trigger.new[i]; 
// 挿入されたばかりの見積もり、または、Primary ではなかった見積もりを、 

// Primary にする場合 
if (nw.primary__c && (Trigger.isInsert || !Trigger.old[i].primary__c)) { 

// この見積もりを持つ商談に対して 

// 主見積もりになろうとしている見積もりがすでにある場合は 
if (opWithPrimary.containsKey(nw.opportunity__c)) { 

// どちらの見積もりも不正とし、記録する 
badQuotes.add(opWithPrimary.get(nw.opportunity__c)); 
badQuotes.add(nw.Id); 

} else { 

// この見積もりは、今のところ問題がないので、リストに追加する 
primaries.add(nw); 
 
// ほかの見積もりも同じ商談の主見積もりになろうとしている場合がありうるので、 
// 商談を記録しておく 
opWithPrimary.put(nw.opportunity__c, nw.Id); 

} 
} 

} 
// 不正な見積もりをマークする 
for (Id bad : badQuotes) { 

Quote__c q = Trigger.newMap.get(bad); 
q.primary__c.AddError('More than one quote is attempting to be primary 

for this opportunity: ' + q.opportunity__c); 
} 
 
// 問題のない見積もりの集合から不正な見積もりを取り除く 
// まず ID を集合にコピーすることで、重複を除いてから 
// 不正な見積もりの ID を削除する 
Set<Id> primaryIds = new Set<Id>(); 
for (Quote__c q : primaries) 

primaryIds.add(q.Id); 
 

primaryIds.removeAll(badQuotes); 
 
// これで、重複がなく、問題のない見積もりの ID の集合ができた。 
// これを使って、対応する Quote__c オブジェクトを取得する 
primaries.clear(); 
util.opportunityIds.clear(); 
for (Id primId : primaryIds) { 

Quote__c q = Trigger.newMap.get(primId); 
primaries.add(q); 
// 新たに主見積もりを設定しようとしている商談の ID を 
// トリガを超えて保存しておく必要がある。 
// そこで、before update トリガは主見積もりがないというエラーを発生させない 
util.opportunityIds.add(q.opportunity__c); 

} 
 
// util クラスの getOtherPrimaryQuotes() を呼び出し、 
// その結果を見積もりのリストに置く 
Quote__c[] others = util.getOtherPrimaryQuotes(primaries); 
 
// そして、その中に含まれる見積もりの Primary フラグを偽に設定し、 
// リストを更新する 
for (quote__c q : others) { 

q.primary__c = false; 
} 
try { 

update others; 
} finally { 

util.opportunityIds.clear(); 
} 

} 
} 

util.apex 

 
global class util { 

// トリガ間で通信するためのクラス変数を定義する 
public static Boolean inQuoteTrigger = false; 
public static Set<Id> opportunityIds = new Set<Id>(); 
// 引数にある見積もりを除いて、 
// 引数にある見積もりに関連するすべての商談に対する主見積もりを返す。 
// 引数にある見積もり以外に主見積もりがない場合は null を返す 
public static Quote__c[] getOtherPrimaryQuotes(list<Quote__c > qs) { 

// 引数にある見積もりが参照する商談すべてのリストを取得する 
List<Id> ops = new List<Id>(); 
for(Quote__c q : qs) 
ops.add(q.opportunity__c); 
 

// 引数にある見積もりを除いて、 
// その商談を参照するすべての見積もりを取得する 
Quote__c[] others = [select id from quote__c where opportunity__c in :ops and 

id not in :qs and primary__c = true]; 
return others; 

} 
// メソッド：商談商品を、指定された見積もり q に移動する 
public static void copyLineItemsFromOpp(List<Quote__c> qs) { 

// まず、見積もりのコピーを作成する。これは、最後に見積もりの totals を変更するが、 
// after トリガでは Trigger.new のレコードを変更できないから。 
// リストに deepClone メソッドを適用しているのは、sobject への 
// clone メソッドはシャローコピーとなり、やはり読み取り専用になってしまうため 

List<Quote__c> quoteList = qs.deepClone(true); 
 
try { 

// inQuoteTrigger クラス変数を設定し、 
// quoteLine トリガに対して、この操作中は動かないように指示する 
// （quoteLine トリガについては、課題 3 で説明する） 
inQuoteTrigger = true; 
 
// 次に、見積もり品目の空の配列を作成する 
List<QuoteLine__c> quoteItemList = new List<QuoteLine__c>(); 
 
// 次に、商談ごとに、その商談商品を取得する。 
// 以下のコードは、親に対する関連性を保ったまま子を取得しているが、 
// これは集合関係クエリの手法の一つ。 

// この場合、OpportunityLineItem を直接照会することも可能だが、 
// そうすると、このコードの中で、取得したアイテムをそれぞれの親（商談）に 
// 関連づける必要がある。後で各商談（とその商談商品）からすぐに見積もりがわかるように、 
// マップを作成しておく 
Map<Id, Opportunity> opportunityMap = new Map<Id, Opportunity>( 
[select Id, (select pricebookentry.product2id, color__c, 

unitprice, quantity, totalprice 
from opportunitylineitems order by createddate) from Opportunity 
where Id IN :getOpportunityIds(quoteList)]); 

// 一括処理のため、見積もりに関して反復して処理する 
for(Quote__c quote:quoteList) { 
 

// 見積もりの総額を計算する 

// SOQL クエリを行って、見積もりに関連づけられた商談に対応する商談商品から 
// データを選び出し、返ってきた OpportunityLineItem レコードの配列について 
// 反復処理し、見積もりを集計する 
Double tot = 0; // 集計中の総額 
Integer i = 0; // 品目数 
 
// マップからそれぞれの商談とその商談商品を取得 
Opportunity opp = opportunityMap.get(quote.opportunity__c); 
 
// 見積もりが商談に関連づけられていないときは、スキップする 
if(opp == null) continue; 
 
for (opportunitylineitem oppline : opp.OpportunityLineItems) { 

i++; 
// 各商談商品につき、 
// 総額を計算し 
if (oppline.unitprice != null && oppline.quantity != null) 

tot += oppline.unitprice * oppline.quantity; 
// 新しい見積もり品目を quoteitems 配列に追加し、 

// その値を初期化する 
quoteItemList.add(new quoteline__c( 

quote__c = quote.id, 
// カウンタを使って品目の番号付けを行っている 
// 上の SOQL クエリでは ORDER BY 句によって、 

// その順番どおりに番号付けされる 
name = String.valueOf(i), 
 
// 次に、商談商品の 
// product、unit、price、quantity、total price、color 
// をコピーする 
product__c = oppline.pricebookentry.product2id, 
unitprice__c = oppline.unitprice, 
quantity__c = oppline.quantity, 
totalprice__c = oppline.totalprice, 
color__c = oppline.color__c)); 

} 
// 見積もりの総額を設定する 
quote.total__c = tot; 

          } 
// 見積もり品目の配列全体を 
// 1 つの insert 文でデータベースに追加する 
insert quoteItemList; 
update quoteList; 
 

} finally { 
// 作業領域をクリーンアップするために、inQuoteTrigger を偽にリセット 
inQuoteTrigger = false; 

} 
  } 
 
// 引数のリストに含まれている全見積もりに対応する 
// 商談 ID の重複のない集合を生成する 
private static Set<Id> getOpportunityIds(List<Quote__c> quotes) { 

Set<Id> idSet = new Set<Id>(); 
for(Quote__c quote:quotes) { 
if(quote.opportunity__c != null) idSet.add(quote.opportunity__c); 

} 
return idSet; 
  } 
 
// 品目の色を、マスタ商品カタログの色に設定する。 
// これは変更されている場合がある 
static void updateColors(Id qId) { 

// 最初に品目をメモリに読み込む 
QuoteLine__c[] lineItems = [select product__c from quoteLine__c 
where quote__c = :qId]; 
// product2 ID を見つける 
Set<Id> productIds = new Set<Id>(); 
for (QuoteLine__c line : lineItems) productIds.add(line.product__c); 
// ここで、すぐにこれらの商品すべてに対してクエリを実行する。  
// ID は常に暗黙的に選択されている。 
// IN を使用することに注意 
Map<Id, Product2> products = new Map<Id, Product2>( 

[select color__c from product2 where id in :productIds]); 
// 上記で作成したマップを使用して、品目を更新する 
QuoteLine__c[] lineItemsToUpdate = new QuoteLine__c[0]; 
for (QuoteLine__c line : lineItems) { 
Product2 product = products.get(line.product__c); 
if (product != null && line.color__c != product.color__c) { 

line.color__c = product.color__c; 
lineItemsToUpdate.add(line); 

} 
} 
update lineItemsToUpdate; 

  } 
 
webService static Proposal__c makeProposal(String id) { 

Proposal__c p = new Proposal__c(name = 'propose it', quote__c = id); 
insert p; 
return p; 

} 
 
static testMethod void testQuoteApp() { 

// まず、サンプルアプリケーションで、OpportunityLineItems をテストするのに 
// 必要なレコード Pricebook2、Product2、PricebookEntry を作成する。 
// 最初に標準価格の商品を作成する。 
Product2 p = new product2(name='x', color__c='c'); 
insert p; 
Pricebook2 stdPb = [select Id from Pricebook2 where isStandard=true limit 1]; 
insert new PricebookEntry(pricebook2id = stdPb.id, product2id = p.id, 

unitprice=1.0, isActive=true); 
// 次に、この商品の項目を含む新しい価格表を作成する 
Pricebook2 pb = new pricebook2(name='test'); 
insert pb; 
PricebookEntry pbe = new PricebookEntry(pricebook2id=pb.id, product2id=p.id, 

unitprice=1.0, isActive=true); 
insert pbe; 
// サンプルアプリケーションで 1 番目の例をテストするため、 
// テストメソッドで価格表の項目を使用して商談品目を作成し、 
// 次に、レコード挿入後に商品から 
// Color フィールドが正しくコピーされたことを確認する 
Opportunity o = new Opportunity(name='test', pricebook2id=pb.id, 
stageName='Open', CloseDate=Date.newInstance(2006,10,10)); 

insert o; 
OpportunityLineItem oli = new OpportunityLineItem(opportunityid=o.id, 

pricebookentryid=pbe.id, unitprice=1.5, quantity=2); 
insert oli; 
System.assertEquals('c', [select color__c from opportunitylineitem where id  

=:oli.id].color__c); 
// サンプルアプリケーションの 5 番目の例をテストするため、 

// 主見積もりを作成し、Primary フィールドが非選択でないことを 
// 確認する。コードでテスト用の try...catch を使用して、 
// エラーメッセージが正しいことを確認する 
Quote__c q = new Quote__c(opportunity__c = o.id, primary__c = true); 
insert q; 
// 非主見積もりとマークできない場合 
q.primary__c = false; 
try { 

update q; 
} catch (System.DmlException e) { 

System.assert(e.getMessage().contains('first error: 
FIELD_CUSTOM_VALIDATION_EXCEPTION, Primary quote cannot be marked  

non-primary: 
[Primary__c]'), e.getMessage()); 
System.assertEquals('Primary quote cannot be marked non-primary', 

e.getDmlMessage(0)); 
 
System.assertEquals('Primary__c', e.getDmlFields(0)[0]); 

} 
// サンプルアプリケーションで 2 番目と 3 番目の例をテストするため、 
// 挿入された見積もりに商談商品から正しい見積もりデータ行がコピーされ、 
// 見積もりの合計値が正しく計算されていることを 
// テストメソッドで確認する。 
System.assertEquals(1, [select count() from quoteline__c where quote__c 

= :q.id]); 
// 見積もりの合計は正しい 

System.assertEquals(3.0, [select total__c from quote__c where id  
= :q.id].total__c); 
// サンプルアプリケーションで 4 番目の例をテストするため、 
// 主見積もりの場合は商談も見積もりも削除できないことを、 
// テストメソッドで確認する。 
try { 
delete o; 
System.assert(false); 

} catch (System.DmlException e) { 
System.assert(e.getDmlMessage(0).indexOf('Cannot delete opportunity with a 

quote') > -1); 
} 
try { 

delete q; 
System.assert(false); 

} catch (System.DmlException e) { 
System.assert(e.getDmlMessage(0).indexOf('Primary quote cannot be deleted') > -1); 
 
} 
// 今度はテストメソッドで 2 番目の見積もりを作成し、これを主見積もりにして 
// 1 番目の見積もりが、主見積もりではなくなっていることを確認する 
Quote__c q2 = new Quote__c(opportunity__c = o.id, primary__c = true); 
insert q2; 
System.assert(![select primary__c from quote__c where id = :q.id].primary__c); 

// 同じ挿入および更新の同じ商談に対して、 
// 主見積もりとしてマークされた同じ商談を持つことはできない 
Quote__c[] quotes = new Quote__c[] { new Quote__c(opportunity__c = o.id, 

primary__c = true), new Quote__c(opportunity__c = o.id, 
primary__c = true)} ; 

try { 
insert quotes; 
System.assert(false); 

} catch (System.DmlException e) { 
System.assert(e.getDmlMessage(0).indexOf('More than one quote is attempting 

to be primary for this opportunity: ' + o.id) > -1); 
} 
// 最後に、テストメソッドで見積もり品目に対する DML 操作を実行し、 
// 見積もり品目の値が関連する値の変更に応じて正しく再計算されることを確認する 
 
QuoteLine__c qLine = new QuoteLine__c(quote__c = q.id, product__c = p.id, 

unitprice__c = 2.0, quantity__c = 2); 
QuoteLine__c qLine1 = qLine.clone(); 
insert qLine; 
System.assertEquals(7.0, [select total__c from quote__c where id  
= :q.id].total__c); 
qLine.unitprice__c = 3.0; 
update qLine; 
System.assertEquals(9.0, [select total__c from quote__c where id  
= :q.id].total__c); 
delete qLine; 
System.assertEquals(3.0, [select total__c from quote__c where id  
= :q.id].total__c); 
// 一括品目操作をテストする 
insert new QuoteLine__C[]{qLine1, qLine1.clone()}; 
System.assertEquals(11.0, [select total__c from quote__c where id  
= :q.id].total__c); 

 
// 商品のマスタカラーの更新、次に見積もりのカラーの更新が正しく動作することをテストする 
p.color__c = 'c2'; 
update p; 
updateColors(q.id); 
for (QuoteLine__c line : [select color__c from quoteLine__c where quote__c  

= :q.id]) 
  { 

System.assertEquals('c2', line.color__c); 
       } 
    } 
} 

